


Modelling Using Axisymmetry

Introduction

This tutorial was completed using ANSYS 7.0 This tutorial is intended to outline the steps required to create an axisymmetric model.

The model will be that of a closed tube made from steel. Point loads will be applied at the center of the top and bottom plate to make an analytical verification simple to calculate. A 3/4 cross section view of the tube is shown below.

As a warning, point loads will create discontinuities in the your model near the point of application. If you chose to use these types of loads in your own modelling, be very careful and be sure to understand the theory of how the FEA package is applying the load and the assumption it is making. In this case, we will only be concerned about the stress distribution far from the point of application, so the discontinuities will have a negligible effect.



Preprocessing: Defining the Problem

1. **Give example a Title**
Utility Menu > File > Change Title ...
`/title, Axisymmetric Tube`
2. **Open preprocessor menu**
ANSYS Main Menu > Preprocessor
`/PREP7`
3. **Create Areas**
Preprocessor > Modeling > Create > Areas > Rectangle > By Dimensions
`RECTNG, X1, X2, Y1, Y2`

For an axisymmetric problem, ANSYS will rotate the area around the y-axis at $x=0$. Therefore, to create the geometry mentioned above, we must define a U-shape.

We are going to define 3 overlapping rectangles as defined in the following table:

Rectangle	X1	X2	Y1	Y2
1	0	20	0	5
2	15	20	0	100
3	0	20	95	100

4. **Add Areas Together**

Preprocessor > Modeling > Operate > Booleans > Add > Areas
AADD, ALL

Click the Pick All button to create a single area.

5. **Define the Type of Element**

Preprocessor > Element Type > Add/Edit/Delete...


For this problem we will use the PLANE2 (Structural, Solid, Triangle 6node) element. This element has 2 degrees of freedom (translation along the X and Y axes).

Many elements support axisymmetry, however if the Ansys Elements Reference (which can be found in the help file) does not discuss axisymmetric applications for a particular element type, axisymmetry is not supported.

6. **Turn on Axisymmetry**

While the Element Types window is still open, click the **Options...** button.

Under Element behavior K3 select **Axisymmetric**.



7. **Define Element Material Properties**

Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic

In the window that appears, enter the following geometric properties for steel:

- i. Young's modulus EX: 200000
- ii. Poisson's Ratio PRXY: 0.3

8. **Define Mesh Size**


Preprocessor > Meshing > Size Cntrls > ManualSize > Areas > All Areas

For this example we will use an element edge length of 2mm.

9. **Mesh the frame**

Preprocessor > Meshing > Mesh > Areas > Free > click 'Pick All'

Your model should now look like this:



Solution Phase: Assigning Loads and Solving

1. **Define Analysis Type**


Solution > Analysis Type > New Analysis > Static

ANTYPE, 0

2. **Apply Constraints**

o Solution > Define Loads > Apply > Structural > Displacement > Symmetry B.C. > On Lines

Pick the two edges on the left, at $x=0$, as shown below. By using the symmetry B.C. command, ANSYS automatically calculates which DOF's should be constrained for the line of symmetry. Since the element we are using only has 2 DOF's per node, we could have constrained the lines in the x-direction to create the symmetric boundary conditions.



o Utility Menu > Select > Entities

Select **Nodes** and **By Location** from the scroll down menus. Click **Y coordinates** and type **50** into the input box as shown below, then click OK.




Solution > Define Loads > Apply > Structural > Displacement > On Nodes > Pick All
 Constrain the nodes in the y-direction (UY). This is required to constrain the model in space, otherwise it would be free to float up or down. The location to constrain the model in the y-direction (y=50) was chosen because it is along a symmetry plane. Therefore, these nodes won't move in the y-direction according to theory.

3. Utility Menu > Select > Entities

In the select entities window, click **Sele All** to reselect all nodes. It is important to always reselect all entities once you've finished to ensure future commands are applied to the whole model and not just a few entities. Once you've clicked Sele All, click on **Cancel** to close the window.

4. **Apply Loads**

- Solution > Define Loads > Apply > Structural > Force/Moment > On Keypoints
 Pick the top left corner of the area and click OK. Apply a load of 100 in the FY direction.
- Solution > Define Loads > Apply > Structural > Force/Moment > On Keypoints
 Pick the bottom left corner of the area and click OK. Apply a load of -100 in the FY direction.
- The applied loads and constraints should now appear as shown in the figure below.



5. **Solve the System**
 Solution > Solve > Current LS
 SOLVE

Postprocessing: Viewing the Results

1. **Hand Calculations**

Hand calculations were performed to verify the solution found using ANSYS:
 The stress across the thickness at $y = 50\text{mm}$ is 0.182 MPa .

$$\begin{aligned}\sigma &= \frac{F}{A} = \frac{F}{\pi(r_o^2 - r_i^2)} \\ &= \frac{100\text{ N}}{\pi(20\text{mm}^2 - 15\text{mm}^2)} \\ &= 0.182\text{ MPa}\end{aligned}$$

2. **Determine the Stress Through the Thickness of the Tube**

- Utility Menu > Select > Entities...

Select **Nodes > By Location > Y coordinates** and type **45,55** in the Min,Max box, as shown below and click OK.



- General Postproc > List Results > Nodal Solution > Stress > Components SCOMP
The following list should pop up.


THE FOLLOWING X,Y,Z VALUES ARE IN GLOBAL COORDINATES						
NODE	SX	SY	SZ	SXY	SVZ	SXZ
42	-0.76099E-05	0.17893	0.18640E-04	0.13372E-04	0.0000	0.0000
44	0.13620E-05	0.17873	-0.52933E-03	0.32806E-06	0.0000	0.0000
46	0.60838E-05	0.17867	-0.79076E-03	0.17733E-05	0.0000	0.0000
48	0.53370E-05	0.17866	-0.79294E-03	0.50904E-06	0.0000	0.0000
50	0.13402E-05	0.17872	-0.53104E-03	0.38945E-06	0.0000	0.0000
52	0.76065E-05	0.17893	0.17546E-04	0.13278E-04	0.0000	0.0000
182	0.45984E-04	0.18521	0.23216E-02	0.22292E-04	0.0000	0.0000
184	0.32178E-04	0.18544	0.16776E-02	0.35547E-05	0.0000	0.0000
186	0.24572E-04	0.18551	0.13437E-02	0.11754E-05	0.0000	0.0000
188	0.25668E-04	0.18552	0.13462E-02	0.33412E-06	0.0000	0.0000
190	0.32273E-04	0.18545	0.16803E-02	0.36055E-05	0.0000	0.0000
192	0.46064E-04	0.18522	0.23258E-02	0.22180E-04	0.0000	0.0000
723	0.96010E-04	0.18205	0.74230E-03	0.16929E-03	0.0000	0.0000
724	0.90010E-04	0.18206	0.30056E-03	0.54607E-04	0.0000	0.0000
725	0.87627E-04	0.18206	0.15398E-03	0.14236E-05	0.0000	0.0000
726	0.89870E-04	0.18206	0.30047E-03	0.55975E-04	0.0000	0.0000
727	0.95814E-04	0.18205	0.74157E-03	0.16999E-03	0.0000	0.0000
MINIMUM VALUES						
NODE	42	48	48	727	42	42
VALUE	-0.76099E-05	0.17866	-0.79294E-03	0.16999E-03	0.0000	0.0000

- If you take the average of the stress in the y-direction over the thickness of the tube, $(0.18552 + 0.17866)/2$, the stress in the tube is 0.182 MPa, matching the analytical solution. The average is used because in the analytical case, it is assumed the stress is evenly distributed across the thickness. This is only true when the location is far from any stress concentrators, such as corners. Thus, to approximate the analytical solution, we must average the stress over the thickness.

3. Plotting the Elements as Axisymmetric

Utility Menu > PlotCtrls > Style > Symmetry Expansion > 2-D Axi-symmetric...

The following window will appear. By clicking on **3/4 expansion** you can produce the figure shown at the beginning of this tutorial.



4. Extra Exercise

It is educational to repeat this tutorial, but leave out the key option which enables axisymmetric modelling. The rest of the commands remain the same. If this is done, the model is a flat, rectangular plate, with a rectangular hole in the middle. Both the stress distribution and deformed shape change drastically, as expected due to the change in geometry. Thus, when using axisymmetry be sure to verify the solutions you get are reasonable to ensure the model is intact axisymmetric.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to '**File > Read input from...**' and select the file.

```
finish
/clear

/title, Axisymmetric Tube
/prep7
/triad,off           ! Turns off origin triad marker

rectng,0,20,0,5     ! Create 3 overlapping rectangles
rectng,15,20,0,100
rectng,0,20,95,100

aadd,all            ! Add the areas together

et,1,plane2        ! Define element type
keyopt,1,3,1       ! Turns on axisymmetry
mp,ex,1,200000     ! Young's Modulus
mp,prxy,1,0.3      ! Poisson's ratio

esize,2            ! Mesh size
amesh,all          ! Mesh the area

finish
/solu

V antype,0         ! Static analysis

lsel,s,loc,x,0     ! Select the lines at x=0
```

```
dl,all,,symm          ! Symmetry constraints
lselect,all           ! Re-select all lines

nselect,s,loc,y,50    ! Node select at y=50
d,all,uy,0            ! Constrain motion in y
nselect,all           ! Re-select all nodes

fk,1,fy,-100         ! Apply point loads in center
fk,12,fy,100

solve
finish

/post1
nselect,s,loc,y,45,55 ! Select nodes from y=45 to y=55
prnsol,s,comp         ! List stresses on those nodes
nselect,all           ! Re-select all nodes

/expand,27,axis,,,10 ! Expand the axisymmetric elements
/view,1,1,2,3        ! Change the viewing angle
/replot
```

Reference

<http://www.mece.ualberta.ca/tutorials/ansys>